

lrtest — Likelihood-ratio test of linear hypothesis

Syntax Stored results	Menu Also see	Description	Remarks and examples
--	--	-----------------------------	--------------------------------------

Syntax

```
{sem|gsem} ... , ... (fit model 1)
```

```
estimates store modelname1
```

```
{sem|gsem} ... , ... (fit model 2)
```

```
estimates store modelname2
```

```
lrtest modelname1 modelname2 (compare models)
```

where one of the models is constrained and the other is unconstrained. `lrtest` counts parameters to determine which model is constrained and which is unconstrained, so it does not matter which model is which.

Warning concerning use with `sem`: Do not omit variables, observed or latent, from the model. Constrain their coefficients to be 0 instead. The models being compared must contain the same set of variables. This is because the standard SEM likelihood value is a function of the variables appearing in the model. Despite this fact, use of `lrtest` is appropriate under the relaxed conditional normality assumption.

Note concerning `gsem`: The above warning does not apply to `gsem` just as it does not apply to other Stata estimation commands. Whether you omit variables or constrain coefficients to 0, results will be the same. The generalized SEM likelihood is conditional on the exogenous variables.

Menu

Statistics > SEM (structural equation modeling) > Testing and CIs > Likelihood-ratio test

Description

`lrtest` is a postestimation command for use after `sem`, `gsem`, and other Stata estimation commands.

`lrtest` performs a likelihood-ratio test comparing models. See [\[R\] lrtest](#).

Remarks and examples

[stata.com](#)

See [\[SEM\] example 10](#) and [\[SEM\] example 39g](#).

When using `lrtest` after `sem`, you must be careful that the models being compared have the same observed and latent variables. For instance, the following is allowed:

```
. sem (L1 -> x1 x2 x3) (L1 <- x4 x5) (x1 <- x4) (x2 <- x5)
. estimates store m1
. sem (L1 -> x1 x2 x3) (L1 <- x4 x5)
. estimates store m2
. lrtest m1 m2
```

This is allowed because both models contain the same variables, namely, L1, x1, . . . , x5, even though the second model omitted some paths.

The following would produce invalid results:

```
. sem (L1 -> x1 x2 x3) (L1 <- x4 x5) (x1 <- x4) (x2 <- x5)
. estimates store m1
. sem (L1 -> x1 x2 x3) (L1 <- x4)
. estimates store m2
. lrtest m1 m2
```

It produces invalid results because the second model does not include variable x5 and the first model does. To run this test correctly, you type

```
. sem (L1 -> x1 x2 x3) (L1 <- x4 x5) (x1 <- x4) (x2 <- x5)
. estimates store m1
. sem (L1 -> x1 x2 x3) (L1 <- x4 x5@0)
. estimates store m2
. lrtest m1 m2
```

If we were using `gsem` rather than `sem`, all the above would still be valid.

Stored results

See *Stored results* in [R] [lrtest](#).

Also see

[SEM] [example 10](#) — MIMIC model

[SEM] [example 39g](#) — Three-level model (multilevel, generalized response)

[R] [lrtest](#) — Likelihood-ratio test after estimation

[SEM] [test](#) — Wald test of linear hypotheses

[SEM] [estat stdize](#) — Test standardized parameters

[SEM] [estat eqtest](#) — Equation-level test that all coefficients are zero